Simo K. Kivelä, 13.7.2004

Ääriarvoprobleema

Seuraava tehtävä on vuoden 2000 pitkän matematiikan ylioppilaskokeesta.

> Astia on kärjellään seisova avonainen ympyräkartio. Kartion pohjan säde on 6,6 cm ja sivujana 11,0 cm. Astia on täynnä vettä. Astiaan asetetaan pallo, joka sivuaa kartion vaippaa. Määritä pallon säde siten, että astiasta valuva vesimäärä on mahdollisimman suuri.

Ratkaisu laskentaohjelmalla ei periaatteessa eroa kynällä ja paperilla tehdystä ratkaisusta. Eroja kuitenkin on siinä, mihin ratkaisijan on kiinnitettävä huomiota: ohjelmalla laskettassa mekaaniset laskut ovat suhteellisen ongelmattomia, mutta tehtävän rakenne on syytä hahmottaa selkeästi.

Ratkaisu

Numeroarvoista huolimatta laskennassa kannattaa käyttää symboleja mahdollisimman pitkälle. Merkitään seuraavasti:

Kartion pohjan säde r, sivujanan pituus s, korkeus h. Pallon säde R, astiassa olevan pallosegmentin korkeus x, etäisyys kartion kärjestä pallon ja kartion sivuamispisteeseen y.

Symboleilla on seuraavat numeeriset arvot:

```
In[1]:= numarvot = {r → 6.6, s → 11.0, h → Sqrt[11.0^2 - 6.6^2]}
```

Out[1]= $\{r \to 6.6,\ s \to 11.,\ h \to 8.8\}$

Yhdenmuotoisista kolmioista saadaan seuraavat riippuvuudet:

```
In[2]:= pallonSade = First[Solve[r / s == R / (h - x + R), R]]
```

Out[2]= $\left\{R \to \frac{-h\, r + r\, x}{r - s}\right\}$

```
In[3]:= sivuamisEtaisyys = First[Solve[R / y == r / h, y]] /. pallonSade
```

Out[3]= $\left\{y \to \frac{h\,(-h\, r + r\, x)}{r\,(r - s)}\right\}$

Mallikuvio helpottaa yhdenmuotoisten kolmioiden hahmottamista:

In[4]:=
```
Show[
 Graphics[{Line[{{-r, h}, {0, 0}, {r, h}, {-r, h}}], Line[{{0, 0}, {0, Max[h,
       h - x + R]}}], Line[{{0, h - x + R}, y / s {-r, h}}], PointSize[0.02],
     Point[{0, h - x + R}], Circle[{0, h - x + R}, R]} /. pallonSade /.
    sivuamisEtaisyys /. {x → 4.5} //. numarvot], AspectRatio → Automatic]
```



Out[4]= - Graphics -

Astian sisään jäävän pallosegementin tilavuus segmentin korkeuden funktiona on

In[5]:=
```
segmVol = Pi x^2 (R - x / 3) /. pallonSade
```

Out[5]= $\pi x^2 \left(-\frac{x}{3} + \frac{-h r + r x}{r - s}\right)$

Lausekkeen pätevyyttä rajoittaa toisaalta tilanne, missä pallo sivuaa kartiota sen pohjaympyrää pitkin, ja toisaalta tilanne, missä pallo kokonaan sijaitsee kartion sisällä. Näistä saadaan muuttujalle x seuraavat rajat:

In[6]:=
```
x1 = First[Solve[s == (y /. sivuamisEtaisyys), x]] // Simplify
```

Out[6]= $\left\{x \to \frac{h^2 + (r - s) s}{h}\right\}$

In[7]:=
```
x2 = First[Solve[x == (2 R /. pallonSade), x]] // Simplify
```

Out[7]= $\left\{x \to \frac{2 h r}{r + s}\right\}$

Astiasta poistuva vesimäärä on mahdollisimman suuri, kun kartion sissän jäävän pallosegementin tilavuus on maksimissa. Paikallisten äärarvojen löytämiseksi lasketaan segmentin tilavuuden derivaatta ja sen nollakohdat:

In[8]:=
```
derSegmVol = D[segmVol, x]
```

Out[8]= $\pi \left(-\frac{1}{3} + \frac{r}{r - s}\right) x^2 + 2 \pi x \left(-\frac{x}{3} + \frac{-h r + r x}{r - s}\right)$

In[9]:= `nollakohdat = Solve[derSegmVol == 0, x]`

Out[9]= $\left\{\{x \to 0\}, \left\{x \to \frac{2\,h\,r}{2\,r + s}\right\}\right\}$

Tässä vaiheessa on luonnollista laskea numeeriset arvot sekä tarkasteluvälille että derivaatan nollakohdille:

In[10]:= `numVali = x /. {x1, x2} /. numarvot`

Out[10]= `{3.3, 6.6}`

In[11]:= `numNollat = x /. nollakohdat /. numarvot`

Out[11]= `{0, 4.8}`

Jälkimmäinen nollakohta siis osuu tarkasteluvälille. Maksimiarvot saadaan joko välin päätepisteissä tai derivaatan nollakohdissa. Selkeintä on piirtää funktion kuvaaja:

In[12]:= `Plot[segmVol /. numarvot, {x, numVali[[1]], numVali[[2]]}]`



Out[12]= `- Graphics -`

Maksimiarvo siis todellakin saadaan, kun segmentin korkeus on $x = 4.8$ cm. Vastaava pallon säde, pallon tilavuus ja segmentin tilavuus ovat

In[13]:= `{R, 4 / 3 Pi R^3, segmVol} /. pallonSade /. nollakohdat[[2]] /. numarvot`

Out[13]= `{6., 904.779, 318.482}`

Harjoitustehtäviä

1) Piirrä kuvio tilanteesta, missä pallosegementin tilavuus on suurimmillaan. Piirrä kuviot tarkasteluvälin päätepisteitä vastaavasta tilanteesta.

2) Suorita laskut uudelleen, kun kartion kartion pohjan säde on 5 cm ja sivujanan pituus 20 cm.

3) Tutki, sijaitseeko derivaatan toinen nollakohta aina tarkasteluvälillä riippumatta lukuarvoista, ts. kartion muodosta.

4) Ratkaise tehtävä uudelleen valitsemalla muuttujaksi kartion kärjen ja pallon keskipisteen etäisyys z segmentin korkeuden sijasta. Piirrä myös syrjäytetyn vesimäärän kuvaaja tämän etäisyyden funktiona, kun $0 \leq z < \infty$.